
and instruction, the result is 0. True and true should produce true, not false. 
Although you can account for situations like this, it usually requires a few 
extra instructions and is somewhat less efficient when computing boolean 
operations.

A system that uses nonzero values to represent true and 0 to represent 
false is an arithmetic logical system. A system that uses the two distinct values like 
0 and 1 to represent false and true is called a boolean logical system, or simply a 
boolean system. You can use either system, as convenient. Consider again 
the boolean expression

     b := ((x = y) and (a <= d)) or ((z - a) <> 5);

The simple expressions resulting from this expression might be:

          mov( x, eax );
          cmp( y, eax );
          sete( al );       // al := x = y;

          mov( a, ebx );
          cmp( ebx, d );
          setle( bl );     // bl := a <= d;
          and( al, bl );   // bl := (x = y) and (a <= d);

          mov( z, eax );
          sub( a, eax );
          cmp( eax, 5 );
          setne( al );
          or( bl, al );     // al := ((x = y) and (a <= d)) or ((z - a) <> 5);
          mov( al, b );

When working with boolean expressions don’t forget that you might be 
able to optimize your code by simplifying those boolean expressions. You can 
use algebraic transformations to help reduce the complexity of an expression. 
In the chapter on control structures, you’ll also see how to use control flow 
to calculate a boolean result. This is generally quite a bit more efficient than 
using complete boolean evaluation as the examples in this section teach.

6.4 Machine and Arithmetic Idioms

An idiom is an idiosyncrasy. Several arithmetic operations and 80x86 instruc-
tions have idiosyncrasies that you can take advantage of when writing assembly 
language code. Some people refer to the use of machine and arithmetic idioms 
as “tricky programming” that you should always avoid in well-written programs. 
While it is wise to avoid tricks just for the sake of tricks, many machine and 
arithmetic idioms are well known and commonly found in assembly language 
programs. Some of them are little more than tricks, but a good number of 
them are simply “tricks of the trade.” This text cannot even begin to present 
all of the idioms in common use today; they are too numerous and the list is 
constantly changing. Nevertheless, there are some very important idioms that 
you will see all the time, so it makes sense to discuss those.

6.4.1 Multiplying without mul, imul, or intmul

When multiplying by a constant, you can sometimes write faster code by using 
shifts, additions, and subtractions in place of multiplication instructions.

Remember, a shl instruction computes the same result as multiplying the 
specified operand by 2. Shifting to the left two bit positions multiplies the 
operand by 4. Shifting to the left three bit positions multiplies the operand 
by 8. In general, shifting an operand to the left n bits multiplies it by 2n. You 
can multiply any value by some constant using a series of shifts and additions 
or shifts and subtractions. For example, to multiply the AX register by 10, you 
need only multiply it by 8 and then add in two times the original value. That 
is, 10 * ax = 8 * ax + 2 * ax. The code to accomplish this is:

          shl( 1, ax );          // Multiply ax by two.
          mov( ax, bx);          // Save 2*ax for later.
          shl( 2, ax );          // Multiply ax by eight (*4 really, 

// but ax contains *2).
          add( bx, ax );         // Add in ax*2 to ax*8 to get ax*10.

Many x86 processors can multiply the AX register (or just about any reg-
ister, for that matter) by various constant values much faster by using shl than 
by using the mul instruction. This may seem hard to believe because it takes 
only one instruction to compute this product:

          intmul( 10, ax );

However, if you look at the instruction timings, the shift and add example 
above requires fewer clock cycles on many processors in the 80x86 family than 
the mul instruction. Of course, the code is somewhat larger (by a few bytes), 
but the performance improvement is usually worth it.

You can also use subtraction with shifts to perform a multiplication oper-
ation. Consider the following multiplication by 7:

          mov( eax, ebx );             // Save eax * 1
          shl( 3, eax );               // eax = eax * 8
          sub( ebx, eax );             // eax * 8 - eax * 1 is eax * 7

A common error beginning assembly language programmers make is 
subtracting or adding 1 or 2 rather than eax * 1 or eax * 2. The following does 
not compute eax * 7:

          shl( 3, eax );
          sub( 1, eax );

It computes (8 * eax) - 1, something entirely different (unless, of course, 
EAX = 1). Beware of this pitfall when using shifts, additions, and subtractions 
to perform multiplication operations.

You can also use the lea instruction to compute certain products. The 
trick is to use the scaled index addressing modes. The following examples 
demonstrate some simple cases:

          lea( eax, [ecx][ecx] );       // eax := ecx * 2
          lea( eax, [eax][eax*2] );     // eax := eax * 3
          lea( eax, [eax*4] );          // eax := eax * 4
          lea( eax, [ebx][ebx*4] );     // eax := ebx * 5
          lea( eax, [eax*8] );          // eax := eax * 8
          lea( eax, [edx][edx*8] );     // eax := edx * 9

6.4.2 Division Without div or idiv

Just as the shl instruction is useful for simulating a multiplication by a power 
of 2, the shr and sar instructions can simulate a division by a power of 2. 
Unfortunately, you cannot easily use shifts, additions, and subtractions to per-
form a division by an arbitrary constant. Therefore, keep in mind that this 
trick is useful only when dividing by powers of 2. Also, don’t forget that the sar 
instruction rounds towards negative infinity rather than toward 0; this is not 
the way the idiv instruction operates (it rounds toward 0).

Another way to perform division is to use the multiply instructions. You 
can divide by some value by multiplying by its reciprocal. Because the multiply 
instruction is faster than the divide instruction, multiplying by a reciprocal is 
usually faster than division.

Now you’re probably wondering, “How does one multiply by a reciprocal 
when the values we’re dealing with are all integers?” The answer, of course, is 
that we must cheat to do this. If you want to multiply by 1/10, there is no way 
you can load the value 1/10 into an 80x86 integer register prior to perform-
ing the multiplication. However, we could multiply 1/10 by 10, perform the 
multiplication, and then divide the result by 10 to get the final result. Of 
course, this wouldn’t buy you anything; in fact, it would make things worse 
because you’re now doing a multiplication by 10 as well as a division by 10. 
However, suppose you multiply 1/10 by 65,536 (6,553), perform the multi-
plication, and then divide by 65,536. This would still perform the correct 
operation, and, as it turns out, if you set up the problem correctly, you can get 
the division operation for free. Consider the following code that divides AX 
by 10:

          mov( 6554, dx );          // 6,554 = round( 65,536/10 )
          mul( dx, ax );

This code leaves AX/10 in the DX register.
To understand how this works, consider what happens when you multiply 

AX by 65,536 ($1_0000). This simply moves AX into DX and sets AX to 0 (a 
multiply by $1_0000 is equivalent to a shift left by 16 bits). Multiplying by 6,554 
(65,536 divided by 10) puts AX divided by 10 into the DX register. Because 
mul is faster than div, this technique runs a little faster than using a division.

Multiplying by a reciprocal works well when you need to divide by a con-
stant. You could even use it to divide by a variable, but the overhead to compute 
the reciprocal pays off only if you perform the division many, many times 
(by the same value).

6.4.3 Implementing Modulo-N Counters with and

If you want to implement a counter variable that counts up to 2n − 1 and then 
resets to 0, simply use the following code:

          inc( CounterVar );
          and( nBits, CounterVar );

where nBits is a binary value containing n bits containing ones right justified 
in the number. For example, to create a counter that cycles between 0 and 15 
(24 − 1), you could use the following:

          inc( CounterVar );
          and( %00001111, CounterVar );

6.5 Floating-Point Arithmetic

When the 8086 CPU first appeared in the late 1970s, semiconductor technology 
was not to the point where Intel could put floating-point instructions directly 
on the 8086 CPU. Therefore, Intel devised a scheme whereby it could use a 
second chip to perform the floating-point calculations—the floating-point 
unit (or FPU).1 By the release of the Intel Pentium chip, semiconductor tech-
nology had advanced to the point that the FPU was fully integrated onto the 
80x86 CPU. Therefore, almost all modern 80x86 CPU devices fully support 
floating-point arithmetic directly on the CPU.

6.5.1 FPU Registers

The 80x86 FPUs add 13 registers to the 80x86: eight floating-point data reg-
isters, a control register, a status register, a tag register, an instruction pointer, 
and a data pointer. The data registers are similar to the 80x86’s general-purpose 
register set insofar as all floating-point calculations take place in these registers. 
The control register contains bits that let you decide how the FPU handles 
certain degenerate cases like rounding of inaccurate computations; it also 
contains bits that control precision and so on. The status register is similar to 
the 80x86’s flags register; it contains the condition code bits and several other 
floating-point flags that describe the state of the FPU. The tag register con-
tains several groups of bits that determine the state of the value in each of the 
eight floating-point data registers. The instruction and data pointer registers 

1 Intel has also referred to this device as the Numeric Data Processor (NDP), Numeric Processor 
Extension (NPX), and math coprocessor.

